



DECUS

PROGRAM LIBRARY

DECUS NO.

8-95

TITLE

TRACE for EAE

AUTHOR

Eberhard Werner

COMPANY

Scripps Institution of Oceanography
San Diego, California

DATE

August 5, 1967

SOURCE LANGUAGE

ATTENTION

This is a USER program. Other than requiring that it conform to submittal and review standards, no quality control has been imposed upon this program by DECUS.

The DECUS Program Library is a clearing house only; it does not generate or test programs. No warranty, express or implied, is made by the contributor, Digital Equipment Computer Users Society or Digital Equipment Corporation as to the accuracy or functioning of the program or related material, and no responsibility is assumed by these parties in connection therewith.

2. ABSTRACT

TRACE INTERPRETIVELY EXECUTES A PDP-8 PROGRAM. AT THE SAME TIME A PRINTOUT IS PROVIDED OF THE CONTENTS OF THE PROGRAM COUNTER, THE INSTRUCTION, THE LINK, ACCUMULATOR, AND MULTIPLIER-QUOTIENT REGISTERS, AND WHERE APPLICABLE, THE EFFECTIVE ADDRESS, AND THE CONTENTS OF THE EFFECTIVE ADDRESS. THIS PRINTOUT MAY BE FOR ALL INSTRUCTIONS OR FOR A SELECTED TYPE OF INSTRUCTION WITHIN SELECTED MEMORY BOUNDS. THE PROGRAM IS CAPABLE OF HANDLING ALL PDP-8 INSTRUCTIONS INCLUDING IOT, TWO-WORD EAE, AND INTERRUPT INSTRUCTIONS. TRACE CANNOT BE DESTROYED BY THE PROGRAM BEING TRACED.

3. REQUIREMENTS

3.1. TRACE REQUIRES TWO AND ONE-HALF PAGES OF MEMORY. THE HALF PAGE MAY BE DELETED IF NO INTERRUPT PROCESSING IS NEEDED. THE PROGRAM MAY BE ASSEMBLED FOR LOCATION IN ANY AREA OF MEMORY. HOWEVER, IF THE PROGRAM IS TO BE LOCATED IN NON-CONSECUTIVE PAGES THE SELF-PROTECT FEATURE MUST BE DISABLED (SEE SECTION 4.5 - 4.6).

3.3. OUTPUT IS ON THE ASR-33 TELETYPE. THE PROGRAM USES THE TYPE 182 EXTENDED ARITHMETIC ELEMENT.

4. USAGE

4.3. INPUTS TO TRACE ARE BY SWITCH SETTINGS. SEE SEC. 4.4.

4.4. (A) START THE PROGRAM AT EITHER OF ITS STARTING ADDRESSES (SEE (E) BELOW). COMPUTER WILL HALT.
(B) ENTER THE INSTRUCTION MASK IN THE SR.
(C) PRESS CONTINUE.
(D) REPEAT (B) AND (C) FOR THE DETERMINING BITS OF THE INSTRUCTION, THE LOWER BOUND - 1 FOR TYPEOUT, THE UPPER BOUND FOR TYPEOUT, AND THE STARTING ADDRESS OF THE PROGRAM TO BE TRACED.
(E) STARTING AT "BEGIN" WILL CLEAR THE PSEUDO-LINK, PSEUDO-ACCUMULATOR, AND THE LINE COUNTER IN THE PSEUDO-ROUTINE. STARTING AT "BEGIN"+1 WILL SAVE THE PREVIOUS CONTENTS OF THE THREE REGISTERS (I. E., STARTING AT "BEGIN" SIMULATES "START", AND STARTING AT "BEGIN"+1 SIMULATES "CONTINUE").

4.5. (A) EXECUTION OF A ICF (6042) INSTRUCTION WILL HANG UP THE PROGRAM.
(B) ANY MEMORY REFERENCE INSTRUCTION WHICH REFERENCES AN ADDRESS WITHIN THE BOUNDS OF TRACE IS NOT EXECUTED. COMPUTER WILL HALT AT "BEGIN"+175.

4.6. (A) RECOVERY FROM HANGUP IS TO PROCEED PAST THE WAITING LOOP. SET "BEGIN"+61 INTO THE PROGRAM COUNTER. PRESS CONTINUE.
(B) IF THE INSTRUCTION CAN BE IGNORED, SIMPLY PRESS CONTINUE. THE INSTRUCTION WILL BE TREATED AS A "NOP" AND THE PROGRAM CONTINUES. IF TRACE HAS BEEN ASSEMBLED FOR NON-CONSECUTIVE PAGES AND THE REFERENCE IS TO A LOCATION BETWEEN THE PAGES, THE SELF-PROTECT FEATURE MAY BE DISABLED BY PLACING A "NOP" IN LOCATION "CONT".

5. RESTRICTIONS (NONE)

6. DESCRIPTION

6.1. TRACE IS A GENERAL DEBUGGING PROGRAM. IT WILL PROVIDE A LISTING OF PROGRAM STEPS AS THEY ARE EXECUTED. THIS IS PARTICULARLY USEFUL IN DEBUGGING PROGRAMS WHICH CONTAIN ERRORS WHICH CAUSE THESE PROGRAMS TO DESTROY THEMSELVES. A PRINTOUT IS PROVIDED OF THE CONTENTS OF THE PROGRAM COUNTER, THE INSTRUCTION, THE CONTENTS OF THE LINK, ACCUMULATOR, AND MULTIPLIER QUOTIENT REGISTERS, AND, WHERE APPLICABLE, THE EFFECTIVE ADDRESS, AND THE CONTENTS OF THE EFFECTIVE ADDRESS. THE TYPE OF INSTRUCTION AND/OR THE SECTION OF MEMORY TO BE DOCUMENTED MAY BE SELECTED WHEN THE PROGRAM IS STARTED. THE INSTRUCTION MASK ENTERED WILL DETERMINE WHICH BITS ARE TO BE INSPECTED, AND THE DETERMINING BITS WILL SPECIFY THE STATE REQUIRED. THIS ALLOWS FLEXIBILITY AND ALLOWS DEBUGGING TO BE DONE FASTER BY DUMPING ONLY INSTRUCTIONS WHICH ARE SUSPECTED OF CAUSING TROUBLE OR WHICH WILL SHOW THE PATH OF THE PROGRAM. ALSO IF THE AREA OF THE PROGRAM WHICH IS CAUSING TROUBLE IS KNOWN, THE MEMORY BOUNDS FOR THE TYPEOUT MAY BE SELECTED TO GIVE TYPEOUT ONLY WHEN THE PROGRAM IS IN THE TROUBLE AREA. IT IS POSSIBLE TO STOP TRACE AND RESTART WITH DIFFERENT MASKS AND BOUNDS BY SIMPLY STOPPING THE PROGRAM WHILE A LINE IS BEING TYPED, RESTARTING AT "BEGIN"+1 (BY PRESSING CONTINUE AFTER SETTING THE PC), AND SETTING THE NEW MASKS AND BOUNDS AS IN SECTION 4.4. THE PROGRAM IS SELF-PROTECTING. ANY ATTEMPT TO REFERENCE A LOCATION WITHIN TRACE IS IGNORED (SEE SECTIONS 4.5(B) - 4.6(B)). OTHERWISE ALL INSTRUCTIONS MAY BE EXECUTED (SEE SECTION 4.5(A) FOR POSSIBLE EXCEPTION). HALTS AND INTERRUPTS ARE GIVEN SPECIAL TREATMENT HALTS CAUSE TYPING OF THE ADDRESS AT WHICH THEY OCCUR AND THE MESSAGE "HLT". THE PROGRAM COUNTER SHOWS "BEGIN"+341 WHEN THE INSTRUCTION IS EXECUTED AND COMPUTER HALTS. THE PSEUDO-PC, I. E., THE ADDRESS OF THE INSTRUCTION WHICH WAS ABOUT TO BE EXECUTED, FOLLOWED BY THE MESSAGE "INT". SEE EXAMPLE 3, SECTION 6.2 FOR EXAMPLES OF BOTH HALTS AND INTERRUPTS.

6.2. EXAMPLES.

THE MOST COMMON USES OF THIS PROGRAM ARE TO: (1) LIST ALL PROGRAM STEPS, OR (2) LIST ALL JUMPS.

EXAMPLE 1. TO LIST ALL PROGRAM STEPS.

- (A) START PROGRAM.
- (B) ENTER 0000 IN SR. CONTINUE.
- (C) CONTINUE.
- (D) CONTINUE.
- (E) ENTER 7777 IN SR. CONTINUE.
- (F) ENTER SA OF PROGRAM TO BE TRACED IN SR. CONTINUE.

THE FIRST SEVERAL LINES OF A TRACE OF THE TRACE PROGRAM STARTING AT 200 ARE SHOWN BELOW.

0200	5231	00000	0000	0231	3722
0231	3722	00000	0000	0560	0000
0232	3626	00000	0000	0556	0000
0233	7040	00000	0000		
0234	1310	07777	0000	0310	7716
0235	3306	07715	0000	0306	7716
0236	5201	00000	0000	0201	7602
0201	7602	00000	0000		
0201	HLT				
0202	7406	00000	0000		
0202	HLT				
0203	3623	00000	0000	0562	0000
0204	7406	00000	0000		
0204	HLT				
0205	0623	00000	0000	0562	0000
0206	7041	00000	0000		
0207	3624	10000	0000	0561	0000
0210	7406	10000	0000		
0210	HLT				
0211	7040	10000	0000		
0212	3227	17777	0000	0227	0000
0213	7406	10000	0000		
0213	HLT				
0214	7040	17777	0000		
0215	3230	10000	0000	0230	0000
0216	7404	10000	0000		
0217	3625	11000	0000	0554	0000
0220	6046	10000	0000		
0221	5622	10000	0000	0401	4772
0401	4772	10000	0000	0615	0000
0616	1254	10000	0000	0654	0001
0617	7640	10001	0000		
0620	5251	10000	0000	0651	1255

EXAMPLE 2. TO LIST ALL JUMPS (JMP AND JMS).

(A) START PROGRAM.

(B) ENTER 6000 IN SR. CONTINUE. (JUMP INSTRUCTIONS ARE DETERMINED BY THE STATE OF THE FIRST TWO BITS.)

(C) ENTER 4000 IN SR. CONTINUE. BIT 0 OF A JUMP INSTRUCTION IS ON, BIT 1 IS OFF.

(D) ENTER 0000 IN SR. CONTINUE.

(E) SEE STEPS (E) AND (F) IN EXAMPLE 1.

THE FIRST FEW LINES OF THE SAME PROGRAM AS ABOVE ARE SHOWN ON THE NEXT PAGE. THE SAME PORTION AS ABOVE IS SHOWN.

0200	5231	00000	0000	0231	3722
0236	5201	00000	0000	0201	7602
0201	HLT				
0202	HLT				
0204	HLT				
0210	HLT				
0213	HLT				
0221	5622	10000	0000	0401	4772
0401	4772	10000	0000	0615	0402
0620	5251	10000	0000	0651	1255

EXAMPLE 3. ILLUSTRATION OF INTERRUPT PROCESSING.

INTERRUPT TESTING IS DONE INTERPRETIVELY. ONLY SPECIFIC FLAGS ARE TESTED. THIS IS DONE IN SUBROUTINE "INT", AND FLAG TESTS FOR ANY DESIRED DEVICES CAN BE KEYED INTO THE APPROPRIATE LOCATIONS OR INSERTED IN THE SOURCE PROGRAM IN THIS SUBROUTINE.

1000	6026	00000	0000		
1001	6001	00000	0000		
1002	5202	00000	0000	1002	5202
1002	INT				
0001	7402	00000	0000		
0001	HLT				

7. METHODS

7.1. EACH INSTRUCTION IS EXAMINED IN TURN AND GROUPED AS AN IOT OR OPERATE INSTRUCTION OR AS A MEMORY REFERENCE INSTRUCTION. THE ADDRESS REFERENCED BY MEMORY REFERENCE INSTRUCTIONS IS DETERMINED AND STORED IN A SPECIFIC MEMORY LOCATION (ADDR). THE ADDRESS IS THEN TESTED TO CHECK IF IT IS WITHIN THE BOUNDS OF TRACE. IF SO, THE EXECUTION IS DELETED, AND THE COMPUTER HALTS. RESTARTING FROM HERE STARTS THE PROCEDURE OVER WITH THE NEXT INSTRUCTION IN THE PROGRAM BEING TRACED.

IF THE INSTRUCTION REFERENCES A LEGAL ADDRESS IT IS NEXT TESTED FOR A JUMP (JMP OR JMS). IF THE OPERATION IS NOT A JUMP THE INSTRUCTION IS ASSEMBLED AS X I ADDR WHERE X IS THE OPERATION AND ADDR IS THE LOCATION OF THE ADDRESS REFERENCED. THE INSTRUCTION IS THEN EXECUTED. JUMP INSTRUCTIONS ARE EXECUTED INTERPRETIVELY. OPERATE AND IOT INSTRUCTIONS ARE TESTED FOR HALT INSTRUCTIONS, INTERRUPT INSTRUCTIONS, AND TWO-WORD EAE INSTRUCTIONS. THESE THREE TYPES OF INSTRUCTIONS MUST BE SPECIALLY HANDLED. ALL OTHERS ARE DIRECTLY EXECUTED. INTERRUPT INSTRUCTIONS SET THE PSEUDO-ION FLAG. HALT INSTRUCTIONS CAUSE TYPING OUT OF THE MESSAGE XXXX HLT, WHERE XXXX IS THE LOCATION OF THE HALT INSTRUCTION, BEFORE EXECUTION. FOR TWO-WORD EAE INSTRUCTIONS THE SECOND WORD IS FETCHED AND THE INSTRUCTION THEN EXECUTED.

8. FORMAT.

8.3. OUTPUT FROM TRACE IS ONE LINE PER INSTRUCTION, 50 LINES PER PAGE. ALL VALUES ARE IN OCTAL DIGITS AND ARE AS FOLLOWS: PROGRAM COUNTER, INSTRUCTION, LINK AND ACCUMULATOR, MULTIPLIER-QUOTIENT REGISTER, ADDRESS REFERENCED, AND CONTENTS OF SAME. THE LAST TWO ARE APPLICABLE ONLY TO MEMORY REFERENCE INSTRUCTIONS. SEE EXAMPLE BELOW.

PC	INST	L AC	MQ	ADDR (ADDR)
1001	6001	00000	0000	
1002	5202	00000	0000	1002 5202

9. EXECUTION TIME

THE PROGRAM IS USUALLY LIMITED BY THE SPEED OF THE TELETYPE. SINCE EACH LINE CONSISTS OF EITHER 27 OR 39 CHARACTERS THIS MEANS AN EXECUTION TIME OF 2.7 OR 3.9 SECONDS PER INSTRUCTION. HOWEVER, IF NO TYPING IS DONE, EXECUTION TIME IS ABOUT 50 TIMES THE NORMAL PROGRAM EXECUTION TIME.

10. PROGRAM.

10.4. PROGRAM LISTING. SEE FOLLOWING PAGES FOR A PASS THREE LISTING FROM PAL III OF TRACE ASSEMBLED TO START AT LOCATION 200.

11. DIAGRAMS (NONE)

12. REFERENCES (NONE)

13. ADDENDA

POSSIBLE MODIFICATIONS.

(A) IF THE PROGRAM TO BE TRACED HAS NO "ION" INSTRUCTION, THE FOLLOWING CHANGES MAY BE MADE.

- (1) "NOP" TO LOCATION BEGIN+201.
- (2) DELETE BEGIN+306.
- (3) DELETE BEGIN+374 AND BEGIN+375.
- (4) DELETE 3RD PAGE.

(B) IF TYPEOUT OF SOME REGISTERS IS NOT REQUIRED THE APPROPRIATE PAIR OF INSTRUCTIONS MAY BE DELETED OR ALTERNATIVELY, THE "JMS I OCT" MAY BE REPLACED BY "CLA".

(C) OUTPUT ON A DEVICE OTHER THAN THE TELETYPE WILL REQUIRE CHANGING LOCATIONS BEGIN+20, +-57, +-61, +-62, AND BEGIN+263.

/PROGRAM TRACE VERSION 5/VIII/67 - EWW
 /TRACE - PAGE 1
 /SETUP AND SUBROUTINES

0200	5231	BEGIN,	JMP SETUP
0201	7602		HLT CLA
0202	7406		OSR HLT /SET INSTRUCTION MASK
0203	3623		DCA I MSK
0204	7406		OSR HLT /SET INSTRUCTION BITS
0205	0623		AND I MSK
0206	7041		CMA IAC
0207	3624		DCA I SET
0210	7406		OSR HLT /SET LOWER BOUNDS
0211	7040		CMA
0212	3227		DCA LBOUND
0213	7406		OSR HLT /SET UPPER BOUND
0214	7040		CMA
0215	3230		DCA UBOUND
0216	7404		OSR /SET STARTING ADDRESS
0217	3625		DCA I PCI
0220	6046		TLS
0221	5622		JMP I START
0222	0401	START,	RETURN+1
0223	0563	MSK,	MASK
0224	0562	SET,	BITSET
0225	0555	PCI,	PC
0226	0557	LINK,	L
0227	0000	LBOUND,	0
0230	0000	UBOUND,	0
0231	3721	SETUP,	DCA I ACI /CLEAR AC AND LINK (PSEUDO)
0232	3626		DCA I LINK
0233	7040		CMA /RESET PAGE LINE COUNTER
0234	1307		TAD M62
0235	3305		DCA LFC
0236	5201		JMP BEGIN+1
0237	0000	OCTCO,	0 /TYPE FOUR DIGIT OCTAL NUMBER
0240	7421		MQL
0241	1255		TAD M4
0242	3266		DCA CRLF /PUT AWAY DIGIT COUNTER
0243	7413	SHIFT,	SHL
0244	0002		2
0245	4351		JMS OUTPUT
0246	4260		4260
0247	2266		ISZ CRLF
0250	5243		JMP SHIFT
0251	4351		JMS OUTPUT /PUT OUT TWO SPACES
0252	0240		240
0253	4240		4240
0254	5637		JMP I OCTCO

0255	7774	M4,	- 4	
0256	0000	TYPE,	0	/TYPE ONE CHARACTER
0257	6041		TSF	
0260	5257		JMP .-1	
0261	6046		TLS	
0262	6041		TSF	
0263	5262		JMP .-1	
0264	7200		CLA	
0265	5656		JMP I TYPE	
0266	0000	CRLF,	0	/TYPE CARRIAGE RETURN-LINE FEED
0267	4351		JMS OUTPUT	
0270	0215		215	
0271	4212		4212	
0272	2305		ISZ LFC	
0273	5666		JMP I CRLF	
0274	1306	PAGE,	TAD M20	/END OF PAGE, TYPE 16 LINE FEEDS
0275	3305		DCA COUNT	
0276	4351		JMS OUTPUT	
0277	4212		4212	
0300	2305		ISZ COUNT	
0301	5276		JMP .-3	
0302	1307		TAD M62	
0303	3305		DCA LFC	
0304	5666		JMP I CRLF	
		COUNT,		
0305	7716	LFC,	-62	
0306	7760	M20,	-20	
0307	7716	M62,	-62	
0310	0000	HALT,	0	/TYPE MESSAGE "HLT"
0311	4266		JMS CRLF	
0312	1625		TAD I PCI	
0313	4237		JMS OCTCO	
0314	4351		JMS OUTPUT	
0315	0310		310	/H
0316	0314		314	/L
0317	4324		4324	/T
0320	5710		JMP I HALT	
0321	0561	ACI,	AC	
0322	7004	JUMP,	RAL	/JUMP INSTRUCTION
0323	7710		SPA CLA	/RETURN JUMP ?
0324	5333		JMP CHDN	/NO
0325	1750		TAD I ADD	/YES
0326	3351		DCA ADDI	
0327	1625		TAD I PCI	/PUT PC IN JUMP ADDRESS
0330	7001		IAC	
0331	3751		DCA I ADDI	
0332	2750		ISZ I ADD	
0333	1750	CHDN,	TAD I ADD	/PUT JUMP ADDRESS IN PC
0334	3625		DCA I PCI	
0335	5622		JMP I START	

0336	0000	BOUND,	0	/CHECK IF PC WITHIN BOUNDS
0337	7100		CLL	
0340	1625		TAD I PCI	
0341	1227		TAD LBOUND	
0342	7200		CLA	
0343	1625		TAD I PCI	
0344	1230		TAD UBOUND	
0345	7640		SZA CLA	/IN BOUNDS IF OVERFLOW ONLY ONCE
0346	2336		ISZ BOUND	/WITHIN BOUNDS - GO TO CALL + 2
0347	5736		JMP I BOUND	
0350	0420	ADD,	ADDR	
		ADDI,		
		OUTPUT,	0	/OUTPUT CHARACTERS UNTIL ONE WITH
0351	0000		TAD I OUTPUT	/NEGATIVE SIGN IS FOUND
0352	1751		JMS TYPE	
0353	4256		TAD I OUTPUT	
0354	1751		ISZ OUTPUT	
0355	2351		SMA CLA	
0356	7700		JMP OUTPUT+1	
0357	5352		JMP I OUTPUT	
0360	5751		/SELF - PROTECT ROUTINE	
0361	7600	FIRST,	-BEGIN	
0362	7110	LAST,	-FINAL-1	
0363	0000	SELF,	0	
0364	7100		CLL	
0365	1750		TAD I ADD	
0366	1361		TAD FIRST	
0367	7200		CLA	
0370	1750		TAD I ADD	
0371	1362		TAD LAST	
0372	7620		SNL CLA	/OVERFLOW TWICE OR NO OVERFLOW
0373	5763		JMP I SELF	/INDICATES ADDR OUTSIDE PROGRAM
0374	3750		DCA I ADD	/ZERO TO ADDRESS IF INSIDE PROGRAM
0375	7402		HLT	
0376	5777		JMP I RETI	/DO NOT EXECUTE CURRENT INSTRUCTION
0377	0400	RETI,	RETURN	
		LAST01,		

/TRACE PAGE 2 - BASIC INTERPRETIVE ROUTINES

```

                                *BEGIN+200
0400 2355 RETURN, ISZ PC
0401 4773 JMS I INTI
0402 3337 DCA FLAG
0403 1755 TAD I PC /GET INSTRUCTION
0404 3340 DCA INST2
0405 1755 TAD I PC /CHECK IF MEMORY REFERENCE
0406 7144 CMA CLL RAL
0407 7720 SMA SNL CLA
0410 5246 JMP DIR
0411 1755 NON, TAD I PC /MODIFY INST AND
0412 0351 AND INMK /DETERMINE OPERAND
0413 1352 TAD CONST /ADDRESS
0414 3340 DCA INST2
0415 1755 TAD I PC /INCREMENT AUTO-INDEX REGISTER IF
0416 0371 AND ZMSK /INSTRUCTION IS AN INDIRECT
0417 3220 DCA .+1 /REFERENCE TO
0420 7402 ADDR, HLT /ONE OF THESE
0421 7040 CMA
0422 3337 DCA FLAG
0423 1755 TAD I PC
0424 0354 AND AMSK
0425 3220 DCA ADDR
0426 1755 TAD I PC /DETERMINE IF ZERO PAGE ADDRESS
0427 7006 RTL
0430 7006 RTL
0431 7700 SMA CLA
0432 5237 JMP ZER /ZERO PAGE
0433 1355 TAD PC /ADD PAGE NUMBER
0434 0265 AND CMSK /FOR CURRENT PAGE
0435 1220 TAD ADDR
0436 3220 DCA ADDR
0437 1755 ZER, TAD I PC
0440 7006 RTL /DIRECT OR INDIRECT?
0441 7004 RAL
0442 7700 SMA CLA
0443 5246 JMP DIR /DIRECT ADDRESS
0444 1620 TAD I ADDR /INDIRECT ADDRESS
0445 3220 DCA ADDR
0446 4775 DIR, JMS I BOUNDS
0447 5300 JMP CONT
0450 1755 TAD I PC
0451 0363 AND MASK
0452 1362 TAD BITSET
0453 7640 SZA CLA
0454 5300 JMP CONT
0455 4767 JMS I CRLFI
0456 1355 TAD PC /YES, TYPE PROGRAM COUNTER
0457 4756 JMS I OCT

```

0460	1755		TAD I PC	/TYPE INSTRUCTION
0461	4756		JMS I OCT	
0462	1357		TAD L	/LINK
0463	1370		TAD IC260	
0464	6046		TL S	
0465	7600	CMSK,	7600	
0466	1361		TAD AC	/ACCUMULATOR
0467	4756		JMS I OCT	
0470	1353		TAD MQ	/MULTIPLIER-QUOTIENT
0471	4756		JMS I OCT	
0472	2337		ISZ FLAG	
0473	5300		JMP CONT	
0474	1220		TAD ADDR	/TYPE ADDRESS REFERENCED
0475	4756		JMS I OCT	
0476	1620		TAD I ADDR	/AND CONTENTS OF SAME
0477	4756		JMS I OCT	
0500	4776	CONT,	JMS I ISELF	/NO, DON'T TYPE
0501	1755		TAD I PC	/TEST FOR HALT INSTRUCTION
0502	0364		AND HMSK	
0503	1366		TAD IMSK	
0504	7650		SNA CLA	
0505	4760		JMS I HALTI	
0506	4774		JMS I INTERI	/CHECK IF INTERRUPT INSTRUCTION
0507	1755		TAD I PC	/CHECK FOR JUMP INSTRUCTION
0510	7004		RAL	
0511	7020		CML	
0512	7520		SNL SMA	
0513	5772		JMP I JUMPI	
0514	7300		CLA CLL	
0515	1351		TAD INMK	/NOP TO INST2-1
0516	3337		DCA INST2-1	
0517	1755		TAD I PC	
0520	0365		AND SMSK	/EXTENDED ARITHMETIC INST,
0521	1366		TAD IMSK	/OF TWO WORD TYPE ?
0522	7070		CMA CML RAR	
0523	7730		SPA SZL CLA	
0524	5332		JMP HOME	/NO, CONTINUE
0525	1755		TAD I PC	/YES, SET UP BOTH WORDS
0526	3337		DCA INST2-1	
0527	2355		ISZ PC	
0530	1755		TAD I PC	/SECOND WORD
0531	3340		DCA INST2	
0532	1357	HOME,	TAD L	/RESTORE OPERATING REGISTERS
0533	7110		CLL RAR	
0534	1353		TAD MQ	
0535	7421		MO L	
0536	1361		TAD AC	
0537	0000	FLAG,	0	
0540	0000	INST2,	0	/EXECUTE INSTRUCTION
0541	7410		SKP	/IN CASE INSTRUCTION WAS A SKIP

0542	2355		ISZ PC	/THE PSEUDO-PC IS INCREMENTED
0543	3361		DCA AC	/SAVE OPERATING REGISTERS
0544	7004		RAL	
0545	3357		DCA L	
0546	7501		MQA	
0547	3353		DCA MQ	
0550	5200		JMP RETURN	/AND GO BACK FOR NEXT INSTRUCTION
				/TABLE OF VALUES
0551	7000	INMK,	7000	
0552	0620	CONST,	AND I ADDR	
0553	0000	MQ,	0	
0554	0177	AMSK,	177	
0555	0000	PC,	0	
0556	0237	OCT,	OCTCO	
0557	0000	L,	0	
0560	0310	HALTI,	HALT	
0561	0000	AC,	0	
0562	0000	BITSET,	0	
0563	0000	MASK,	0	
0564	7403	HMSK,	7403	
0565	7407	SMSK,	7407	
0566	0376	IMSK,	376	
0567	0266	CRLFI,	CRLF	
0570	0260	IC260,	260	
0571	0777	ZMSK,	777	
0572	0322	JUMPI,	JUMP	
0573	0615	INTI,	INT	
0574	0600	INTERI,	INTER	
0575	0336	BOUNDS,	BOUND	
0576	0363	I SELF,	SELF	
		LAST02,		

/TRACE PAGE 3

```

                                /INTERRUPT TESTING AND PROCESSING
                                *BEGIN+400
0600 0000 INTER, 0 /CHECK FOR INTERRUPT INSTRUCTION
0601 1657 TAD I PCII
0602 3262 DCA IPC
0603 1662 TAD I IPC
0604 0260 AND C7774
0605 1261 TAD M6000
0606 7640 SZA CLA
0607 5600 JMP I INTER
0610 1662 TAD I IPC
0611 0256 AND C3
0612 7110 CLL RAR
0613 3255 DCA INTFL2
0614 5665 JMP I IRET

0615 0000 INT, 0 /INTERRUPT CHECK
0616 1254 TAD INTFLG
0617 7640 SZA CLA /IS INTERRUPT ON?
0620 5251 JMP END /NO
0621 6031 KSF /YES, CHECK VARIOUS FLAGS
0622 7410 SKP
0623 5234 JMP YES
0624 7000 NOP
0625 7410 SKP
0626 5234 JMP YES
0627 6011 RSF
0630 7410 SKP
0631 5234 JMP YES
0632 6021 PSF
0633 5251 JMP END
0634 7001 YES, IAC /SET UP FOR A JMS 0
0635 3255 DCA INTFL2
0636 4664 JMS I CRLI
0637 1657 TAD I PCII
0640 4666 JMS I OCTNO
0641 1657 TAD I PCII
0642 3000 DCA 0
0643 7001 IAC
0644 3657 DCA I PCII
0645 4663 JMS I TYPI
0646 0311 311
0647 0316 316
0650 4324 4324
0651 1255 END, TAD INTFL2
0652 3254 DCA INTFLG
0653 5615 JMP I INT
0654 0001 INTFLG, 1
0655 0001 INTFL2, 1
```

0656	0003	C3,	3
0657	0555	PCII,	PC
0660	7774	C7774,	7774
0661	2000	M6000,	-6000
0662	0000	IPC,	0
0663	0351	TYPI,	OUTPUT
0664	0266	CRLI,	CRLF
0665	0400	IRET,	RETURN
0666	0237	OCTNO,	OCTCO
		FINAL,	

/SYMBOL TABLE FOR TRACE
 /VERSION 5/VIII/67

AC	0561	LAST02	0577
ACI	0321	LBOUND	0227
ADD	0350	LFC	0305
ADDI	0351	LINK	0226
ADDR	0420	MASK	0563
AMSK	0554	MQ	0553
BEGIN	0200	MSK	0223
BITSET	0562	M20	0306
BOUND	0336	M4	0255
BOUNDS	0575	M6000	0661
CHDN	0333	M62	0307
CMASK	0465	NON	0411
CONST	0552	OCT	0556
CONT	0500	OCTCO	0237
COUNT	0305	OCTNO	0666
CRLF	0266	OUTPUT	0351
CRLFI	0567	PAGE	0274
CRLI	0664	PC	0555
C3	0656	PCI	0225
C7774	0660	PCII	0657
DIR	0446	RETURN	0400
END	0651	RETI	0377
FINAL	0667	SELF	0363
FIRST	0361	SET	0224
FLAG	0537	SETUP	0231
HALT	0310	SHIFT	0243
HALTI	0560	SMSK	0565
HMSK	0564	START	0222
HOME	0532	TYPE	0256
IC260	0570	TYPI	0663
IMSK	0566	UBOUND	0230
INMK	0551	YES	0634
INST2	0540	ZER	0437
INT	0615	ZMSK	0571
INTER	0600		
INTERI	0574		
INTFLG	0654		
INTFL2	0655		
INTI	0573		
IPC	0662		
IRET	0665		
ISELF	0576		
JUMP	0322		
JUMPI	0572		
L	0557		
LAST	0362		
LAST01	0400		

